GFUN AND THE AGM

Bruno Salvy

January 1998

Let a and b be two positive real numbers, with $b < a$. The *arithmetic-geometric* mean of a and b is classically defined as the common limit of the sequences a_k and b_k defined by $a_{k+1} = \frac{a_k + b_k}{2}, b_{k+1} = \sqrt{a_k b_k}$, with $a_0 = a$ and $b_0 = b$.

That the sequences converge to the same limit can be inferred from

$$a_{k+1}^2 - b_{k+1}^2 = \left(\frac{a_k - b_k}{2}\right)^2 .$$

This common limit is known by Maple as GaussAGM (a, b). It was discovered by Gauss that the arithmetic-geometric mean is related to hypergeometric functions by

```
> GaussAGM(a,b)=a/hypergeom([1/2, 1/2],[1],1-b^2/a^2);
```

$$\mathrm{GaussAGM}(a, b) = \frac{a}{\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], 1 - \frac{b^2}{a^2}\right)}$$

```
> eval(subs(a=3.,b=2.,"));
```

$$2.474680436 = 2.474680437$$

This worksheet, largely inspired by [1], shows how gfun can be used to guess and then prove this result, as well as a generalization of it due to J. M. Borwein and P. B. Borwein.

The functional equation

Following [1], we start by introducing a generalization of the arithmetic-geometric mean obtained by considering the following iteration where $1 < N$ is an integer:

$$a_{k+1} = \frac{a_k + (N-1) b_k}{N}, b_{k+1} = \left(a_{k+1}^N - \left(\frac{a_k - b_k}{N}\right)^N\right)^{\left(\frac{1}{N}\right)} .$$

where the second equality is motivated by

$$a_{k+1}^N - b_{k+1}^N = \left(\frac{a_k - b_k}{N}\right)^N$$

from which follows that both sequences converge to a common limit, which is denoted by

$M_N(a,b)$. The arithmetic-geometric mean corresponds to the case $N=2$.

The function $M_N(a,b)$ is easily seen to be homogeneous: $M_N(\lambda a, \lambda b) = \lambda M_N(a,b)$, for $0 < \lambda$. Together with the obvious property that $M_N(a_0, b_0) = M_N(a_1, b_1)$, this implies that for x in $(0,1)$,

$$M_N\left(1, (1-x^N)^{\left(\frac{1}{N}\right)}\right) = (1+(N-1)x)\, M_N\left(1, \frac{1-x}{1+(N-1)x}\right) .$$

Defining the function $A_N(x)$ by

$$A_N(x) = \frac{1}{M_N\left(1, (1-x)^{\left(\frac{1}{N}\right)}\right)}$$

the equation above translates into the following *functional equation* for $A_N(x)$:

$$> \quad funeq := (1+(N-1)x)\, A_N(x^N) = A_N\left(1 - \left(\frac{1-x}{1+(N-1)x}\right)^N\right)$$

which plays a central rôle in this worksheet. It is not too difficult to show that $A_N(x)$ is analytic in the neighborhood of the origin and that the functional equation above has a unique analytic solution in this neighborhood.

The quadratic case

This is the case $N=2$ and Gauss's theorem is equivalent to stating that

$$A_2(x) = \mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], x\right) .$$

We now use gfun to first guess and then prove this result. The first step is to use the functional equation to compute a series expansion of $A_2(x)$, then we use this series to guess a possible closed form which turns out to be analytic, then we show that this analytic function does satisfy the functional equation.

Series expansion

Starting from the functional equation,

```
> funeq2:=subs(N=2,A[2]=A,op(1,funeq)-op(2,funeq));
```

$$funeq2 := (1+x)\,A(x^2) - A\left(1 - \frac{(1-x)^2}{(1+x)^2}\right)$$

a series solution is easily obtained by a method of undeterminate coefficients:

```
> sol:=1:
```

```
> for i to 12 do
sol:=sol+x^i*solve(op(1,series(eval(
subs(A=unapply(sol+a*x^i,x),funeq2)),x,i+2)),a) od:sol;
```

$$1 + \frac{1}{4}x + \frac{9}{64}x^2 + \frac{25}{256}x^3 + \frac{1225}{16384}x^4 + \frac{3969}{65536}x^5 + \frac{53361}{1048576}x^6 + \frac{184041}{4194304}x^7 + \frac{41409225}{1073741824}x^8 + \frac{147744025}{4294967296}x^9 + \frac{2133423721}{68719476736}x^{10} + \frac{7775536041}{274877906944}x^{11} + \frac{457028729521}{17592186044416}x^{12}$$

Guessing the solution

From this series, gfun guesses a differential equation which could be satisfied by $A(x)$:

```
> deq:=op(1,gfun[seriestodiffeq](series(sol,x,13),y(x),[ogf]));
```

$$deq := \{y(0) = 1, y(x) + (-4 + 8x)\left(\frac{\partial}{\partial x}y(x)\right) + (-4x + 4x^2)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right), D(y)(0) = \frac{1}{4}\}$$

It turns out that Maple's dsolve function is unable to solve this differential equation:

```
> dsolve(deq,y(x));
```

We then use gfun[diffeqtorec] which deduces from this differential equation the recurrence satisfied by the Taylor coefficients of its solutions:

```
> gfun[diffeqtorec](deq,y(x),u(n));
```

$$\{(1 + 4n + 4n^2)\,u(n) + (-8n - 4 - 4n^2)\,u(n+1), u(0) = 1, u(1) = \frac{1}{4}\}$$

From this first order linear recurrence, a solution is easily found:

```
> rsolve(",u(n));
```

$$\frac{\Gamma\left(n + \frac{1}{2}\right)^2}{\Gamma(n+1)^2\,\pi}$$

hence the sum:

```
> y(x)=sum("*x^n,n=0..infinity);
```

$$y(x) = \mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], x\right)$$

Proving the result of the guess

The proof consists in showing that the function $y(x)$, which is obviously analytic, satisfies the functional equation

```
> subs(A=y,funeq2)=0;
```

$$(1+x)\,y(x^2) - y\left(1 - \frac{(1-x)^2}{(1+x)^2}\right) = 0$$

Our approach consists in using closure properties of solutions of linear differential equations that are implemented in gfun to compute a linear differential equation satisfied by the left-hand side of this equation. The proof then reduces to showing that 0 is the only solution of this differential equation that is compatible with the initial conditions, which are 0 up to a large order by construction of y.

It turns out that this proof can be performed directly from the differential equation, and would apply even if no closed-form had been found.

Given a linear differential equation satisfied by a series $y(x)$, the function gfun[algebraicsubs] computes a linear differential equation satisfied by $y(f(x))$ for any algebraic function, given by a polynomial P such that $P(x, f(x)) = 0$. Thus a differential equation satisfied by $y\left(1 - \frac{(1-x)^2}{(1+x)^2}\right)$ is easily computed from that satisfied by $y(x)$:

```
> deq:=op(select(has,deq,x)):
```

```
> deq1:=gfun[algebraicsubs](deq,numer(y-(1-(1-x)^2/(1+x)^2)),y(x));
```

$$deq1 := (-1+x)\,y(x) + (-x^3 - 3x^2 - x + 1)\left(\frac{\partial}{\partial x}y(x)\right) + (-x^4 - x^3 + x^2 + x)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

Similarly, $y(x^2)$ satisfies

```
> gfun[algebraicsubs](deq,y-x^2,y(x));
```

$$y(x)\,x + (-1 + 3x^2)\left(\frac{\partial}{\partial x}y(x)\right) + (-x + x^3)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

and its product by $-(1+x)$ satisfies

```
> deq2:=gfun[`diffeq*diffeq`](",y(x)+1+x,y(x));
```

$$deq2 := (1-x)\,y(x) + (x+x^3+3x^2-1)\left(\frac{\partial}{\partial x}y(x)\right) + (-x-x^2+x^3+x^4)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

From there, we deduce a differential equation satisfied by the left-hand side of the functional equation when applied to the hypergeometric function we have guessed:

```
> gfun[`diffeq+diffeq`](deq1,deq2,y(x));
```

$$(1-x)\,y(x) + (x+x^3+3x^2-1)\left(\frac{\partial}{\partial x}y(x)\right) + (-x-x^2+x^3+x^4)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

Analytic solutions of this equation have a coefficient sequence which satisfies

```
> gfun[diffeqtorec](",y(x),u(n));
```

$$\{n^2\,u(n) + (2+n^2+4n)\,u(n+1) + (-2n+1-n^2)\,u(n+2) + (-6n-9-n^2)\,u(n+3),$$
$$u(1) = 4\_C_0,\ u(2) = \_C_0,\ u(0) = 4\_C_0\}$$

and thus the first three zeroes of the Taylor expansion of the left-hand side of the functional equation conclude the proof.

The cubic case

It has been discovered by J. M. Borwein and P. B. Borwein that a hypergeometric expression also exists when $N=3$. Again, the same steps as above lead to guessing and then proving the following result by gfun .

Theorem . [Borwein & Borwein 90] *The function* $A_3(x)$ *corresponding to the AGM iteration of order 3 has the following closed form:*

$$A_3(x) = \mathrm{hypergeom}\left(\left[\frac{1}{3},\frac{2}{3}\right],[1],x\right) .$$

Series expansion

We start from the functional equation

```
> funeq3:=subs(N=3,A[3]=A,op(1,funeq)-op(2,funeq));
```

$$funeq3 := (1+2x)\,A(x^3) - A\left(1-\frac{(1-x)^3}{(1+2x)^3}\right)$$

and compute the first terms of the series expansion of the solution:

```
> sol:=1:
```

```
> for i to 12 do
sol:=sol+x^i*solve(op(1,series(eval(
subs(A=unapply(sol+a*x^i,x),funeq3)),x,i+2)),a) od:sol;
```

$$1+\frac{2}{9}x+\frac{10}{81}x^2+\frac{560}{6561}x^3+\frac{3850}{59049}x^4+\frac{28028}{531441}x^5+\frac{1905904}{43046721}x^6+\frac{14780480}{387420489}x^7$$

$$+\frac{116858170}{3486784401}x^{8}+\frac{75957810500}{2541865828329}x^{9}+\frac{616777421260}{22876792454961}x^{10}+\frac{5056555387520}{205891132094649}x^{11}$$
$$+\frac{376081306946800}{16677181699666569}x^{12}$$

Guessing the solution

Again, this is a lucky situation where a differential equation can be guessed:

```
> deq:=op(1,gfun[seriestodiffeq](series(sol,x,13),y(x),[ogf]));
```

$$deq := \left\{D(y)(0)=\frac{2}{9}, y(0)=1, 2\,y(x)+(-9+18x)\left(\frac{\partial}{\partial x}y(x)\right)+(-9x+9x^{2})\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)\right\}$$

From there, we find a closed-form as before

```
> gfun[diffeqtorec](deq,y(x),u(n));
```

$$\left\{u(1)=\frac{2}{9}, u(0)=1, (2+9n+9n^{2})\,u(n)+(-18n-9-9n^{2})\,u(n+1)\right\}$$

```
> rsolve(",u(n));
```

$$\frac{1}{2}\frac{\Gamma\left(n+\frac{2}{3}\right)\Gamma\left(n+\frac{1}{3}\right)\sqrt{3}}{\Gamma(n+1)^{2}\,\pi}$$

```
> y(x)=sum("*x^n,n=0..infinity);
```

$$y(x)=\mathrm{hypergeom}\left(\left[\frac{1}{3},\frac{2}{3}\right],[1],x\right)$$

Proving the result of the guess

The same routine applies:

```
> subs(A=y,funeq3)=0;
```

$$(1+2x)\,y(x^{3})-y\left(1-\frac{(1-x)^{3}}{(1+2x)^{3}}\right)=0$$

```
> deq:=op(select(has,deq,x)):
```

```
> deq1:=gfun[algebraicsubs](deq,numer(y-(1-(1-x)^3/(1+2*x)^3)),y(x));
```

$$deq1 := (2-4x+2x^{2})\,y(x)+(-1+8x^{5}+12x^{4}+4x^{3}+4x^{2})\left(\frac{\partial}{\partial x}y(x)\right)$$

$$+\left(4x^6+4x^5+x^4-4x^3-4x^2-x\right)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

```
> gfun[algebraicsubs](deq,y-x^3,y(x));
```

$$2\,y(x)\,x^2+\left(-1+4x^3\right)\left(\frac{\partial}{\partial x}y(x)\right)+\left(-x+x^4\right)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

```
> deq2:=gfun[`diffeq*diffeq`](",y(x)+1+2*x,y(x));
```

$$deq2 := \left(2-4x+2x^2\right)y(x)+\left(-1+8x^5+12x^4+4x^3+4x^2\right)\left(\frac{\partial}{\partial x}y(x)\right)$$
$$+\left(4x^6+4x^5+x^4-4x^3-4x^2-x\right)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

```
> gfun[`diffeq+diffeq`](deq1,deq2,y(x));
```

$$\left(2-4x+2x^2\right)y(x)+\left(-1+8x^5+12x^4+4x^3+4x^2\right)\left(\frac{\partial}{\partial x}y(x)\right)$$
$$+\left(4x^6+4x^5+x^4-4x^3-4x^2-x\right)\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}y(x)\right)\right)$$

```
> gfun[diffeqtorec](",y(x),u(n));
```

$$\{\left(4n^2+4n\right)u(n)+\left(16n+12+4n^2\right)u(n+1)+\left(n^2+7n+12\right)u(n+2)$$
$$+\left(-16-16n-4n^2\right)u(n+3)+\left(-46-4n^2-28n\right)u(n+4)$$
$$+\left(-10n-25-n^2\right)u(n+5),\ u(2)=0,\ u(4)=\_C_0,\ u(3)=\frac{1}{2}\_C_0,\ u(0)=\frac{9}{4}\_C_0,$$
$$u(1)=\frac{9}{2}\_C_0\}$$

and thus the first five zeroes of the Taylor expansion of the left-hand side of the functional equation conclude the proof.

Conclusion

These results are very good examples of the use of gfun : experiments first lead to conjecture a general form for the solution to a problem and then a completely different process leads to a proof. However, the apparent ease with which the problems treated here are solved using gfun hides the preliminary work which led to the form under which this approach could work. For example this approach does not seem to work for higher values of N , where similar results might exist.

Bibliography

[1] Arithmetic-Goemetric Means Revisited. Jonathan M. Borwein, Petr Lisonek and John A. Macdonald. *MapleTech* , **4-1** , pp. 20-27 (1997).